
1. Overview

This document lists all the essencial changes needed to correct problems in the TTCN implementation of test case 23.1.1 which is part of the LTE_A_PRO test suite in the ‘iwd-TTCN3-B2017-06_D18wk12’ ATS delivery. 
With these changes applied the test case can be demonstrated to run with one or more UEs (see section 4). Execution log files are provided as evidence.

3. Verification Test Summary

Test Case:
23.1.1
Test Group:
\LTE_A_PRO\23\CIoT_Optimization_CP.ttcn
ATS Version:
iwd-TTCN3-B2017-06_D18wk12
System Simulator used:
Anritsu Protocol Conformance Test System ME7834LA




UE used:
Spreadtrum UIX8910
Verification Status:
PASS


4. Corrections required for CIoT TC 23.1.1
Change 1

	Function name
	TC_23_1_1

	Reason for change
	The TC contains timer of 6 minutes and due to this the test guardtimer needs to be extended to at least 12 minutes.

	Summary of change
	Extended guardtimer to 12 minutes.

	TTCN module
	\LTE_A_PRO_Testsuite.ttcn

	MCC160 Comment
	


Before  change

    testcase TC_23_1_1() runs on MTC_LTE system SYSTEM_LTE {

    // @purpose

    //   CIoT / Control Plane MO and MT IP and non-IP Data Transfer / Serving PLMN Rate Control / APN Rate Control

    var EUTRA_PTC        v_EUTRA      := null;

    var IMS_PTC          v_IMS1       := null;

    var IMS_PTC          v_IMS2       := null;

    timer t_GuardTimer := int2float(360);

    v_EUTRA := EUTRA_PTC.create alive;

    f_MTC_ConnectPTCs_LTE(system, v_EUTRA, v_IMS1, v_IMS2);

    v_EUTRA.start(f_TC_23_1_1_EUTRA());

    t_GuardTimer.start;

    f_MTC_MainLoop_LTE(t_GuardTimer);

  }
After  change

    testcase TC_23_1_1() runs on MTC_LTE system SYSTEM_LTE {

    // @purpose

    //   CIoT / Control Plane MO and MT IP and non-IP Data Transfer / Serving PLMN Rate Control / APN Rate Control

    var EUTRA_PTC        v_EUTRA      := null;

    var IMS_PTC          v_IMS1       := null;

    var IMS_PTC          v_IMS2       := null;

    timer t_GuardTimer := int2float(720);
    v_EUTRA := EUTRA_PTC.create alive;

    f_MTC_ConnectPTCs_LTE(system, v_EUTRA, v_IMS1, v_IMS2);

    v_EUTRA.start(f_TC_23_1_1_EUTRA());

    t_GuardTimer.start;

    f_MTC_MainLoop_LTE(t_GuardTimer);

  }
Change 2

	Function name
	f_GetTestcaseAttrib_CIOT_UP

	Reason for change
	This TC does not cover the User plane CIoT features, it is defined as Control Plane TC.

	Summary of change
	Removed TC from list of User Plane TCs.

	TTCN module
	\LTE_A_PRO\common\TestcaseProperties.ttcn

	MCC160 Comment
	


Before  change

     function f_GetTestcaseAttrib_CIOT_UP(charstring p_Testcase) return boolean

  {

    select (p_Testcase) {

      case ("TC_23_1_1") { return true; }
      case ("TC_23_2_1") { return true; }

      case ("TC_23_2_2") { return true; }

      case ("TC_23_2_3") { return true; }

    }

    return false;

  } 
After  change

    function f_GetTestcaseAttrib_CIOT_UP(charstring p_Testcase) return boolean

  {

    select (p_Testcase) {

      case ("TC_23_2_1") { return true; }

      case ("TC_23_2_2") { return true; }

      case ("TC_23_2_3") { return true; }

    }

    return false;

  }  
Change 3

	Function name
	fl_ReturnCIOT_RRCConnectionSetupComplete_v8a0

	Reason for change
	The optional IE ue-CE-NeedULGaps-r13 is not expected in TTCN and needs to be added.
Draft Prose CR for 36.508 will be raised.

	Summary of change
	Optional IE ue-CE-NeedULGaps-r13 is included in RRCConnectionSetupComplete.

	TTCN module
	Common\EUTRA\EUTRA_RRC_Templates.ttcn

	MCC160 Comment
	


Before  change

     function fl_ReturnCIOT_RRCConnectionSetupComplete_v8a0() return template RRCConnectionSetupComplete_v8a0_IEs

  { //@sic R5-172999 sic@

    var template RRCConnectionSetupComplete_v8a0_IEs  v_RRCConnectionSetupComplete_v8a0 := *;

    var template (omit) RRCConnectionSetupComplete_UP_CIoT_Type v_UP_CIoT_Type := omit;

    var template (omit) RRCConnectionSetupComplete_CP_CIoT_Type v_CP_CIoT_Type := omit;

    if (f_GetTestcaseAttrib_CIOT_UP(testcasename()) or f_GetTestcaseAttrib_CIOT_CP(testcasename())) {

      if (pc_User_Plane_CIoT_Optimisation) {

        v_UP_CIoT_Type := true_;

      }

      if (pc_Control_Plane_CIoT_Optimisation) {

        v_CP_CIoT_Type := true_;

      }

      //@sic R5-177082 sic@

      v_RRCConnectionSetupComplete_v8a0 := cr_RRCConnectionSetupComplete_v8a0_IEs(cr_RRCConnectionSetupComplete_v1020_IEs(cr_RRCConnectionSetupComplete_v1130_IEs(cr_RRCConnectionSetupComplete_v1250_IEs(cr_RRCConnectionSetupComplete_v1320_IEs(v_UP_CIoT_Type, v_CP_CIoT_Type)))));  //@sic R5-177082 sic@

    }

    return v_RRCConnectionSetupComplete_v8a0

  }  

After  change

        function fl_ReturnCIOT_RRCConnectionSetupComplete_v8a0() return template RRCConnectionSetupComplete_v8a0_IEs

  { //@sic R5-172999 sic@

    var template RRCConnectionSetupComplete_v8a0_IEs  v_RRCConnectionSetupComplete_v8a0 := *;

    var template (omit) RRCConnectionSetupComplete_UP_CIoT_Type v_UP_CIoT_Type := omit;

    var template (omit) RRCConnectionSetupComplete_CP_CIoT_Type v_CP_CIoT_Type := omit;

    if (f_GetTestcaseAttrib_CIOT_UP(testcasename()) or f_GetTestcaseAttrib_CIOT_CP(testcasename())) {

      if (pc_User_Plane_CIoT_Optimisation) {

        v_UP_CIoT_Type := true_;

      }

      if (pc_Control_Plane_CIoT_Optimisation) {

        v_CP_CIoT_Type := true_;

      }

      //@sic R5-177082 sic@

      v_RRCConnectionSetupComplete_v8a0 := cr_RRCConnectionSetupComplete_v8a0_IEs(cr_RRCConnectionSetupComplete_v1020_IEs(cr_RRCConnectionSetupComplete_v1130_IEs(cr_RRCConnectionSetupComplete_v1250_IEs(cr_RRCConnectionSetupComplete_v1320_IEs(v_UP_CIoT_Type, v_CP_CIoT_Type, cr_RRCConnectionSetupComplete_v1330_IEs)))));  //@sic R5-177082 sic@
    }

    return v_RRCConnectionSetupComplete_v8a0

  }  

Change 4

	Function name
	f_EUTRA_RRCConnectionReconfigurationWithAttachAcceptAllParams

	Reason for change
	As per TS 36.508 Table 4.7.3-6: ACTIVATE DEFAULT EPS BEARER CONTEXT REQUEST:
In case of CIoT_ControlPlane_PDN the Control plane only indication needs to be set to ‘0001’B.

	Summary of change
	Included Control plane only indication in case of CIoT_ControlPlane_PDN.

	TTCN module
	\Common\EUTRA\EUTRA_NASSteps.ttcn

	MCC160 Comment
	


Before  change

  function f_EUTRA_RRCConnectionReconfigurationWithAttachAcceptAllParams(EUTRA_CellId_Type p_CellId,

                                                                         HalfOctet_Type p_EpsBearerId,

                                                                         RRC_TransactionIdentifier p_RRC_TI,

                                                                         ProcedureTransactionIdentifier p_EPS_TI,

                                                                         NAS_AttDetValue_Type p_EpsAttachResultValue,

                                                                         template (omit) PDN_Address p_PDN_Address,

                                                                         template (omit) MobileIdentity p_Guti,

                                                                         template (value) TrackingAreaIdList p_TaiList,

                                                                         template (omit) LocAreaId p_LAI,

                                                                         template (omit) MobileIdentity p_MSId,

                                                                         template (omit) ESM_Cause p_Cause,  // @sic R5-104703 sic@

                                                                         template (value) AccessPointName p_APN,

                                                                         template (omit) ProtocolConfigOptions p_Pco, // @sic R5-104703 sic@ Default value is PPP

                                                                         template (omit) AdditionalUpdateResult p_AdditionalUpdateResult, // @sic R5-110796 sic@

                                                                         template (value) GPRS_Timer p_T3412 := cs_GprsTimer_v_deact, // @sic R5s100780 sic@

                                                                         template (omit) GPRS_Timer p_T3402 := omit, // @sic R5s100780 sic@

                                                                         template (omit) GPRS_Timer p_T3423 := omit, // @sic R5s100780 sic@

                                                                         template (omit) PLMN_List p_EquivalentPlmnList := omit, // @sic R5s100780 sic@

                                                                         template (omit) EmergNumList p_EmergNumList := omit, // @sic R5s100780 sic@

                                                                         template (omit) EPS_NetworkFeatureSupport p_NtwkFeatSupport := f_GetNetworkFeatureSupport(), //  @sic R5s100780, R5-173000 sic@

                                                                         template (omit) GPRS_Timer3 p_T3412Extd := omit,

                                                                         SecurityHeaderType p_SecurityStatus := tsc_SHT_IntegrityProtected_Ciphered,

                                                                         template (omit) BEARER_CONTEXT_TYPE p_BearerContextNumber := omit,

                                                                         //@sic R5-145971 sic@

                                                                         template (omit) EMM_Cause p_EMM_Cause := omit,

                                                                         template (omit) GPRS_Timer2 p_T3324 := omit, // @sic R5s150329 Baseline Moving sic@

                                                                         template (omit) DCN_ID p_DCN_Id := omit)

    runs on EUTRA_PTC

  { /* RRC connection reconfiguration acc. to 36.331 cl. 5.3.5 */

    /* @sic R5-110796 Addition of 9.2.1.2.1d: f_GetAdditionalUpdateResult(p_RcvdAddUpdateType, p_ForcedAttach) replaced by p_AdditionalUpdateResult (p_RcvdAddUpdateType, p_ForcedAttach removed sic@ */

    /* @sic R5-144370: p_APN mandatory; has to be handed over => f_ESM_InitialRegistration_InitAPNandPCOs shall be used to initialise the APN sic@ */

    /* @sic R5-144797: new parameter p_BearerContextNumber sic@ */

    var template (value) NAS_MSG_Request_Type v_NAS_MSG_Request;

    var template (value) AccessPointName v_APN := p_APN;         /* @sic R5-144370: no evaluation of the APN in this function anymore sic@ */

    var DRB_Identity v_DRB_Id := f_EUTRA_NB_EpsBearerAssociatedDRB(p_EpsBearerId);   /* @sic R5-144797 sic@ */

    var template(omit) ServingPLMNRateControl v_ServingPLMNRateControl := fl_AssignServingPLMNRateControl(); //@sic R5-172883 sic@

    v_NAS_MSG_Request := cs_NAS_RequestWithPiggybacking(p_SecurityStatus,

                                                        cs_ATTACH_ACCEPT_Common(p_EpsAttachResultValue,

                                                                                p_T3412,

                                                                                p_TaiList,

                                                                                p_Guti,

                                                                                p_LAI,

                                                                                p_MSId,

                                                                                p_EMM_Cause, // @sic R5-104703 R5-145971 sic@

                                                                                p_T3402,

                                                                                p_T3423,

                                                                                p_EquivalentPlmnList,

                                                                                p_EmergNumList,

                                                                                p_NtwkFeatSupport,  // @sic R5-103674 sic@

                                                                                p_AdditionalUpdateResult,

                                                                                p_T3412Extd, // @sic R5s1120178 Baseline Moving sic@

                                                                                p_T3324, //  @sic R5s150329 Baseline Moving sic@

                                                                                omit,

                                                                                p_DCN_Id), // @sic R5-177082  sic@

                                                        cs_508_ActivateDefEpsBearerContextRequest(p_EpsBearerId,

                                                                                                  p_EPS_TI,

                                                                                                  v_APN,

                                                                                                  p_PDN_Address,

                                                                                                  p_Cause, // @sic R5-104703 sic@

                                                                                                  p_Pco,

                                                                                                  p_BearerContextNumber,

                                                                                                  -,-,-,-,

                                                                                                  v_ServingPLMNRateControl));

    f_IP_ChangeEutraCell(IP, p_CellId);                                 /* @sic R5-113734 change 8 sic@ */

    f_EUTRA_MobileInfo_SetNetworkFeatureSupport(p_NtwkFeatSupport);     /* @sic R5-177127: set global variable acc. to what has been sent to the UE in the ATTACH ACCEPT sic@ */

    f_EUTRA_RRCConnectionReconfiguration_SRB2_1AM_DRB_Est(p_CellId,

                                                          p_RRC_TI,

                                                          v_NAS_MSG_Request,

                                                          v_DRB_Id);

  };
After  change

import from CIOT_NASTemplates all;      
…

function f_EUTRA_RRCConnectionReconfigurationWithAttachAcceptAllParams(EUTRA_CellId_Type p_CellId,

                                                                         HalfOctet_Type p_EpsBearerId,

                                                                         RRC_TransactionIdentifier p_RRC_TI,

                                                                         ProcedureTransactionIdentifier p_EPS_TI,

                                                                         NAS_AttDetValue_Type p_EpsAttachResultValue,

                                                                         template (omit) PDN_Address p_PDN_Address,

                                                                         template (omit) MobileIdentity p_Guti,

                                                                         template (value) TrackingAreaIdList p_TaiList,

                                                                         template (omit) LocAreaId p_LAI,

                                                                         template (omit) MobileIdentity p_MSId,

                                                                         template (omit) ESM_Cause p_Cause,  // @sic R5-104703 sic@

                                                                         template (value) AccessPointName p_APN,

                                                                         template (omit) ProtocolConfigOptions p_Pco, // @sic R5-104703 sic@ Default value is PPP

                                                                         template (omit) AdditionalUpdateResult p_AdditionalUpdateResult, // @sic R5-110796 sic@

                                                                         template (value) GPRS_Timer p_T3412 := cs_GprsTimer_v_deact, // @sic R5s100780 sic@

                                                                         template (omit) GPRS_Timer p_T3402 := omit, // @sic R5s100780 sic@

                                                                         template (omit) GPRS_Timer p_T3423 := omit, // @sic R5s100780 sic@

                                                                         template (omit) PLMN_List p_EquivalentPlmnList := omit, // @sic R5s100780 sic@

                                                                         template (omit) EmergNumList p_EmergNumList := omit, // @sic R5s100780 sic@

                                                                         template (omit) EPS_NetworkFeatureSupport p_NtwkFeatSupport := f_GetNetworkFeatureSupport(), //  @sic R5s100780, R5-173000 sic@

                                                                         template (omit) GPRS_Timer3 p_T3412Extd := omit,

                                                                         SecurityHeaderType p_SecurityStatus := tsc_SHT_IntegrityProtected_Ciphered,

                                                                         template (omit) BEARER_CONTEXT_TYPE p_BearerContextNumber := omit,

                                                                         //@sic R5-145971 sic@

                                                                         template (omit) EMM_Cause p_EMM_Cause := omit,

                                                                         template (omit) GPRS_Timer2 p_T3324 := omit, // @sic R5s150329 Baseline Moving sic@

                                                                         template (omit) DCN_ID p_DCN_Id := omit)

    runs on EUTRA_PTC

  { /* RRC connection reconfiguration acc. to 36.331 cl. 5.3.5 */

    /* @sic R5-110796 Addition of 9.2.1.2.1d: f_GetAdditionalUpdateResult(p_RcvdAddUpdateType, p_ForcedAttach) replaced by p_AdditionalUpdateResult (p_RcvdAddUpdateType, p_ForcedAttach removed sic@ */

    /* @sic R5-144370: p_APN mandatory; has to be handed over => f_ESM_InitialRegistration_InitAPNandPCOs shall be used to initialise the APN sic@ */

    /* @sic R5-144797: new parameter p_BearerContextNumber sic@ */

    var template (value) NAS_MSG_Request_Type v_NAS_MSG_Request;

    var template (value) AccessPointName v_APN := p_APN;         /* @sic R5-144370: no evaluation of the APN in this function anymore sic@ */

    var DRB_Identity v_DRB_Id := f_EUTRA_NB_EpsBearerAssociatedDRB(p_EpsBearerId);   /* @sic R5-144797 sic@ */

    var template(omit) ServingPLMNRateControl v_ServingPLMNRateControl := fl_AssignServingPLMNRateControl(); //@sic R5-172883 sic@

    var template (omit) ControlPlaneInd v_ControlPlaneInd := omit;

    if (f_GetTestcaseAttrib_CIOT_CP(testcasename())) {     

      v_ControlPlaneInd := cs_ControlPlaneInd;

    }
    v_NAS_MSG_Request := cs_NAS_RequestWithPiggybacking(p_SecurityStatus,

                                                        cs_ATTACH_ACCEPT_Common(p_EpsAttachResultValue,

                                                                                p_T3412,

                                                                                p_TaiList,

                                                                                p_Guti,

                                                                                p_LAI,

                                                                                p_MSId,

                                                                                p_EMM_Cause, // @sic R5-104703 R5-145971 sic@

                                                                                p_T3402,

                                                                                p_T3423,

                                                                                p_EquivalentPlmnList,

                                                                                p_EmergNumList,

                                                                                p_NtwkFeatSupport,  // @sic R5-103674 sic@

                                                                                p_AdditionalUpdateResult,

                                                                                p_T3412Extd, // @sic R5s1120178 Baseline Moving sic@

                                                                                p_T3324, //  @sic R5s150329 Baseline Moving sic@

                                                                                omit,

                                                                                p_DCN_Id), // @sic R5-177082  sic@

                                                        cs_508_ActivateDefEpsBearerContextRequest(p_EpsBearerId,

                                                                                                  p_EPS_TI,

                                                                                                  v_APN,

                                                                                                  p_PDN_Address,

                                                                                                  p_Cause, // @sic R5-104703 sic@

                                                                                                  p_Pco,

                                                                                                  p_BearerContextNumber,

                                                                                                  -,

                                                                                                  -,

                                                                                                  v_ControlPlaneInd,
                                                                                                  -,

                                                                                                  v_ServingPLMNRateControl));

    f_IP_ChangeEutraCell(IP, p_CellId);                                 /* @sic R5-113734 change 8 sic@ */

    f_EUTRA_MobileInfo_SetNetworkFeatureSupport(p_NtwkFeatSupport);     /* @sic R5-177127: set global variable acc. to what has been sent to the UE in the ATTACH ACCEPT sic@ */

    f_EUTRA_RRCConnectionReconfiguration_SRB2_1AM_DRB_Est(p_CellId,

                                                          p_RRC_TI,

                                                          v_NAS_MSG_Request,

                                                          v_DRB_Id);

  };

  /*

   * @desc      Return Serving PLMN rate control - Applied to CIOT

   *            Note: for TC 23.1.1

   * @return    template(omit) ServingPLMNRateControl

   * @status    APPROVED (IMS, IMS_IRAT, LTE, LTE_A_IRAT, LTE_A_PRO, LTE_A_R10_R11, LTE_A_R12, LTE_IRAT, POS)

   */

  function fl_AssignServingPLMNRateControl() return template(omit) ServingPLMNRateControl

  {

    var template(omit) ServingPLMNRateControl v_ServingPLMNRateControl := omit;

    if (f_GetTestcaseAttrib_CIOT_CP_PLMN(testcasename())) {

      v_ServingPLMNRateControl := cs_ServingPLMNRateControl(int2oct(10,2)); //@sic R5-174488 sic@

    }

    return v_ServingPLMNRateControl;

  }
Change 5

	Function name
	f_EUTRA_Deactivate_TestMode_SRB1

	Reason for change
	Created a new version of function f_EUTRA_Deactivate_TestMode that uses SRB1 instead of SRB2. SRB2 is not created in the TC.

	Summary of change
	Created a new function that uses SRB1.

	TTCN module
	\Common\EUTRA\EUTRA_LoopBack.ttcn

	MCC160 Comment
	


New function
      function f_EUTRA_Deactivate_TestMode_SRB1(EUTRA_CellId_Type p_CellId) runs on EUTRA_PTC    

  { //@sic R5-090700: TT01 has been removed from 36.509 sic@

    //@sic R5-103847 Deactivate test loop and open test loop is removed sic@

    //@sic R5-113714 re introduces usage of deactivate test loop sic@

    SRB.send (cas_SRB1_NasPdu_REQ(p_CellId,

                                  cs_TimingInfo_Now,

                                  cs_NAS_Request(tsc_SHT_IntegrityProtected_Ciphered,

                                                 cs_DEACTIVATE_TEST_MODE)));

    SRB.receive (car_SRB1_NasPdu_IND(p_CellId,

                                     cr_NAS_Indication(tsc_SHT_IntegrityProtected_Ciphered,

                                                       cr_DEACTIVATE_TEST_MODE_COMPLETE)));

  }
Change 6

	Function name
	f_EUTRA_CloseUE_TestLoopModeGH

	Reason for change
	SRB2 is not created in the TC. Changed to use SRB1.

	Summary of change
	SRB2 changed to SRB1.

	TTCN module
	LTE_A_PRO\23\CIoT_Optimization_CP.ttcn

	MCC160 Comment
	


Before  change

    function f_EUTRA_CloseUE_TestLoopModeGH(EUTRA_CellId_Type p_CellId,

                                          UE_TestLoopMode_Type p_TestLoopMode,

                                          B1_Type p_UplinkMode := tsc_TestLoopModeGH_UplinkMode_RLC,

                                          integer p_Repetitions := 1,

                                          integer p_UplinkDataDelay := 0)

    runs on EUTRA_PTC

  {

    var template (value) UE_TestLoopModeGH_LB_Setup_Type v_TestLoopModeGH_Setup := cs_NBIOT_TestLoopModeGH_LB_Setup(p_UplinkMode, p_Repetitions, p_UplinkDataDelay);

    var template (value) NAS_DL_Message_Type v_CLOSE_UE_TEST_LOOP_Mode := cs_CLOSE_UE_TEST_LOOP_ModeGH(p_TestLoopMode, v_TestLoopModeGH_Setup);

    SRB.send (cas_SRB2_NasPdu_REQ(p_CellId,

                                  cs_TimingInfo_Now,

                                  cs_NAS_Request(tsc_SHT_IntegrityProtected_Ciphered, v_CLOSE_UE_TEST_LOOP_Mode)));

    SRB.receive (car_SRB2_NasPdu_IND(p_CellId,

                                     cr_NAS_Indication(tsc_SHT_IntegrityProtected_Ciphered, cr_CLOSE_UE_TEST_LOOP_COMPLETE)));

  }
After  change

function f_EUTRA_CloseUE_TestLoopModeGH(EUTRA_CellId_Type p_CellId,

                                          UE_TestLoopMode_Type p_TestLoopMode,

                                          B1_Type p_UplinkMode := tsc_TestLoopModeGH_UplinkMode_RLC,

                                          integer p_Repetitions := 1,

                                          integer p_UplinkDataDelay := 0)

    runs on EUTRA_PTC

  {

    var template (value) UE_TestLoopModeGH_LB_Setup_Type v_TestLoopModeGH_Setup := cs_NBIOT_TestLoopModeGH_LB_Setup(p_UplinkMode, p_Repetitions, p_UplinkDataDelay);

    var template (value) NAS_DL_Message_Type v_CLOSE_UE_TEST_LOOP_Mode := cs_CLOSE_UE_TEST_LOOP_ModeGH(p_TestLoopMode, v_TestLoopModeGH_Setup);

    SRB.send (cas_SRB1_NasPdu_REQ(p_CellId,                           

                                  cs_TimingInfo_Now,

                                  cs_NAS_Request(tsc_SHT_IntegrityProtected_Ciphered, v_CLOSE_UE_TEST_LOOP_Mode)));

    SRB.receive (car_SRB1_NasPdu_IND(p_CellId,     
                                     cr_NAS_Indication(tsc_SHT_IntegrityProtected_Ciphered, cr_CLOSE_UE_TEST_LOOP_COMPLETE)));

  }
Change 7
	Function name
	f_EUTRA_RbEst_CP

	Reason for change
	SRB2 is not created in the TC. Changed to use SRB1.

	Summary of change
	SRB2 changed to SRB1.

	TTCN module
	LTE_A_PRO\23\CIoT_Optimization_CP.ttcn

	MCC160 Comment
	


Before  change

     function f_EUTRA_RbEst_CP (EUTRA_CellId_Type p_CellId)

    runs on EUTRA_PTC

  {

    var RRC_TransactionIdentifier v_RRC_TI := tsc_RRC_TI_Def;

    var NAS_MSG_Indication_Type v_NAS_Ind;

    // Step 2 - SS sends a Paging message to the UE on the appropriate paging block, and including the UE identity in one entry of the IE pagingRecordLists.

    f_EUTRA_UE_Page_Def(p_CellId);

    // Steps 3-5

    v_NAS_Ind := f_EUTRA_RRC_ConnEst_DefWithNas(p_CellId,

                                                v_RRC_TI,

                                                cr_EstablishmentCause_Any,

                                                cr_NAS_Indication (tsc_SHT_IntegrityProtected, cr_CONTROL_PLANE_SERVICE_REQUEST(?))

                                               );

    // Step 6 - The SS transmits a SERVICE ACCEPT message.

    SRB.send(cas_SRB2_NasPdu_REQ(eutra_Cell1, cs_TimingInfo_Now, cs_NAS_Request(tsc_SHT_IntegrityProtected_Ciphered, cs_SERVICE_ACCEPT(v_NAS_Ind.Pdu.Msg.cONTROL_PLANE_SERVICE_REQUEST.epsBearerContextStatus))));

  }
After  change

       function f_EUTRA_RbEst_CP (EUTRA_CellId_Type p_CellId)

    runs on EUTRA_PTC

  {

    var RRC_TransactionIdentifier v_RRC_TI := tsc_RRC_TI_Def;

    var NAS_MSG_Indication_Type v_NAS_Ind;

    // Step 2 - SS sends a Paging message to the UE on the appropriate paging block, and including the UE identity in one entry of the IE pagingRecordLists.

    f_EUTRA_UE_Page_Def(p_CellId);

    // Steps 3-5

    v_NAS_Ind := f_EUTRA_RRC_ConnEst_DefWithNas(p_CellId,

                                                v_RRC_TI,

                                                cr_EstablishmentCause_Any,

                                                cr_NAS_Indication (tsc_SHT_IntegrityProtected, cr_CONTROL_PLANE_SERVICE_REQUEST(?))

                                               );

    // Step 6 - The SS transmits a SERVICE ACCEPT message.

    SRB.send(cas_SRB1_NasPdu_REQ(eutra_Cell1, cs_TimingInfo_Now, cs_NAS_Request(tsc_SHT_IntegrityProtected_Ciphered, cs_SERVICE_ACCEPT(v_NAS_Ind.Pdu.Msg.cONTROL_PLANE_SERVICE_REQUEST.epsBearerContextStatus))));
  }
Change 8

	Function name
	f_TC_23_1_1_EUTRA

	Reason for change
	1) As per TC preamble, the UE should be in State 3A-CP which is handled in f_EUTRA_RbEst_CP.
2) The SRB2 is not created in TC body, SRB2 is replaced with SRB1 in all steps.

3) Missing repeat; causes failure as the 2nd ESM DATA TRANSPORT in step 15 is unexpected.
4) f_EUTRA_Deactivate_TestMode is replaced with f_EUTRA_Deactivate_TestMode_SRB1 in step 19-20 as SRB2 is not present.

5) Postamble is modified to not expect only EPS_only detach.

	Summary of change
	1) Used function f_EUTRA_RbEst_CP.
2) SRB2 is replaced with SRB1 in all steps.
3) Included missing repeat;

4) Used f_EUTRA_Deactivate_TestMode_SRB1 in step 19-20.
5) Postamble is modified.

	TTCN module
	LTE_A_PRO\23\CIoT_Optimization_CP.ttcn

	MCC160 Comment
	


Before  change

  function f_TC_23_1_1_EUTRA() runs on EUTRA_PTC

  {/* CIoT / Control Plane MO and MT IP and non-IP Data Transfer / Serving PLMN Rate Control / APN Rate Control */

    var template (value) octetstring v_DataStep3_11 := 'F0F0F0'O;

    var template (value) octetstring v_DataStep17a5 := f_GetN_OctetsFromPRBS(1, 10);

    var EUTRA_SecurityParams_Type v_AuthParams;

    var ProcedureTransactionIdentifier v_PTI := tsc_PTI_Unassigned;

    var integer v_Count := 0;

    var integer v_PCOCount := 0;

    var integer v_DataLengthReceived := 0;

    var integer i;

    var template (value) ProtocolContainer v_ProtocolContainerAPN;

    var template (value) ExtdProtocolConfigOptions v_NewPco;

    var template (value) ProtocolContainer v_ProtocolContainerIPv4;

    var template (value) ProtocolContainer v_ProtocolContainerNonIP;

    var template (value) NAS_ExtdProtocolConfigOptions_Type v_ProtocolContainerList := {};

    var SRB_COMMON_IND v_ReceivedAsp;

    var template ReleaseAssistanceInd v_ReleaseAssistanceInd := cr_ReleaseAssistanceInd ('00'B) ifpresent;

    var template (value) APNRateControlParams v_APN_RateControl;

    timer t_LoopDelay;

    timer t_PLMNRateControl;

    //Initialise all cells, security and mobile parameters

    f_EUTRA_Init(c1);

    //Create and configure Cell 1

    f_EUTRA_CellConfig_Def (eutra_Cell1);

    //UE is in State 3A-CP, Control Plane CioT connection request, UE Test Mode Activated  according to TS 36.508

    //Settings in ACTIVATE DEFAULT EPS BEARER CONTEXT REQUEST, in preamble,  to check PLMN rate control => done in f_EUTRA_RRCConnectionReconfigurationWithAttachAcceptAllParams unsing attribute CIOT_CP_PLMN

    f_EUTRA_Preamble(eutra_Cell1, STATE2A_TESTLOOP_ModeG);

    f_EUTRA_RbEst_Def(eutra_Cell1);

    f_EUTRA_TestBody_Set(true);

    //@siclog "Steps 1 - 2" siclog@

    //The SS transmits a CLOSE UE TEST LOOP message to close the UE test loop mode for user data transfer (12 transmissions; 60 sec delay).

    //The UE transmits a CLOSE UE TEST LOOP COMPLETE message to confirm that loopback is activated.

    f_EUTRA_CloseUE_TestLoopModeGH(eutra_Cell1, tsc_UE_TestLoopMode_TypeG, tsc_TestLoopModeGH_UplinkMode_NAS, 12, 60);

    t_LoopDelay.start( f_EUTRA_SetTimerToleranceMin (eutra_Cell1, nonProtocolTimer, 60.0));

    //@siclog "Step 3" siclog@

    //SS transmits an ESM DATA TRANSPORT message containing downlink user data of 3 octets

    v_AuthParams := f_EUTRA_Security_Get();

    SRB.send(cas_SRB_NasPdu_REQ(eutra_Cell1,

                                tsc_SRB2,

                                cs_TimingInfo_Now,

                                cs_NAS_Request(tsc_SHT_IntegrityProtected_Ciphered,

                                               cs_ESM_DATA_TRANSPORT (tsc_EpsDefaultBearerId, v_PTI, cs_UserDataContainer(v_DataStep3_11)))));

    //@siclog "Step 4" siclog@

    //The SS transmits an RRCConnectionRelease message.

    f_EUTRA_RRC_ConnectionRelease ( eutra_Cell1);

    //@siclog "Step 5" siclog@

    //Wait for the time set in the CLOSE UE TEST LOOP to expire. => 60 seconds

    t_LoopDelay.timeout;

    //@siclog "Step 6" siclog@

    //SS starts timer 6 min PLMN Rate control

    t_PLMNRateControl.start( f_EUTRA_SetTimerToleranceMax (eutra_Cell1, nasTimer, 360.0));

    //@siclog "Step 7" siclog@

    //Check: Does the UE transmit an RRCConnectionRequest message?

    f_EUTRA_RRC_ConnectionRequest_Def(eutra_Cell1);

    //@siclog "Step 8" siclog@

    //SS transmits an RRCConnectionSetup message.

    f_EUTRA_RRC_ConnectionSetup_Def(eutra_Cell1);

    //@siclog "Step 9" siclog@

    //The UE transmits an RRCConnectionSetupComplete message to confirm the successful completion of the connection establishment and to initiate the session management procedure by including the CONTROL PLANE SERVICE REQUEST message.

    f_EUTRA_RRCConnectionSetupComplete_Def(eutra_Cell1,

                                           cr_508_RRCConnectionSetupComplete(tsc_RRC_TI_Def, ?),

                                           cr_NAS_IndicationWithPiggybacking(

                                                tsc_SHT_IntegrityProtected_PartiallyCiphered,

                                                cr_CONTROL_PLANE_SERVICE_REQUEST(v_AuthParams.KSIasme),

                                                cr_ESM_DATA_TRANSPORT(tsc_EpsDefaultBearerId, cr_UserDataContainer(int2oct(lengthof(v_DataStep3_11), 2),v_DataStep3_11))

                                                ));

    f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Step 9");

    //@siclog "Step 10" siclog@

    //The SS transmits a SERVICE ACCEPT message.

     SRB.send(cas_SRB2_NasPdu_REQ(eutra_Cell1, cs_TimingInfo_Now, cs_NAS_Request(tsc_SHT_IntegrityProtected_Ciphered, cs_SERVICE_ACCEPT)));

    //@siclog "Step 11" siclog@

    //Check: Does the UE send an ESM DATA TRANSPORT message containing the same user data sent by the SS in step 3?

    SRB.receive(car_SRB2_NasPdu_IND(eutra_Cell1,

                                    cr_NAS_Indication(tsc_SHT_IntegrityProtected_Ciphered,

                                                      cr_ESM_DATA_TRANSPORT(tsc_EpsDefaultBearerId, cr_UserDataContainer(int2oct(lengthof(v_DataStep3_11),2),v_DataStep3_11)))));

    v_Count := 2;

    f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Step 11");

    //@siclog "Steps 12-16" siclog@

    alt {

      [] SRB.receive(car_SRB2_NasPdu_IND(eutra_Cell1,

                                         cr_NAS_Indication(tsc_SHT_IntegrityProtected_Ciphered,

                                                           cr_ESM_DATA_TRANSPORT(tsc_EpsDefaultBearerId, cr_UserDataContainer(int2oct(lengthof(v_DataStep3_11),2),v_DataStep3_11), v_ReleaseAssistanceInd))))

        {

          v_Count := v_Count +1;

          if (v_Count == 3) {

            f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Step 12");

            repeat;

          } else if (v_Count == 12) {

            f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Step 15");

            //@siclog "Step 16" siclog@

            // SS stops timer 6 min PLMN Rate control

            t_PLMNRateControl.stop;

          } else if (v_Count == 11){

            v_ReleaseAssistanceInd := cr_ReleaseAssistanceInd (('00'B, '01'B)) ifpresent;

          }else {

            repeat;

          }

        }

      [v_Count == 10] t_PLMNRateControl.timeout

        {

           t_PLMNRateControl.start( f_EUTRA_SetTimerToleranceMax (eutra_Cell1, nasTimer, 360.0));

           repeat;

        }

    }

    //Steps 17a1 to 17a8 describe behaviour that depends on UE capabilities; the "lower case letter" identifies a step sequence that take place depending on whether the UE supports APN control

    if (pc_APN_RateControl) {

        //@siclog "Step 17a1" siclog@

        // The SS sends a MODIFY EPS BEARER CONTEXT REQUEST message of to modify APN, MTU and PLMN rate controls.

        v_APN_RateControl := cs_APNRateControlParams ('001'B, int2oct(1, 3));

        v_ProtocolContainerAPN := cs_ProtocolContainer ('0016'O, bit2oct(encvalue(v_APN_RateControl)));

        v_NewPco := f_NAS_ExtdProtocolConfigOptionsTX ({v_ProtocolContainerAPN});

        SRB.send(cas_SRB2_NasPdu_REQ(eutra_Cell1,

                                     cs_TimingInfo_Now,

                                     cs_NAS_Request(tsc_SHT_IntegrityProtected_Ciphered,

                                                    cs_ModifyEPSBearerCxtReq_AllParams(tsc_EpsDefaultBearerId,

                                                                                       v_PTI,

                                                                                       omit,

                                                                                       omit,

                                                                                       omit,

                                                                                       omit,

                                                                                       omit,

                                                                                       f_Get_508_PFI(),

                                                                                       omit,

                                                                                       omit,

                                                                                       omit,

                                                                                       omit,

                                                                                       v_NewPco))));

        //@siclog "Step 17a2" siclog@

        //Check: Does the UE transmit a MODIFY EPS BEARER CONTEXT ACCEPT message?

        SRB.receive(car_SRB2_NasPdu_IND(eutra_Cell1,

                                    cr_NAS_Indication(tsc_SHT_IntegrityProtected_Ciphered,

                                                      cr_508_ModifyEPSBearerCxtAccept(tsc_EpsDedicatedBearerId))));

        f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Step 17a2");

        //@siclog "Steps 17a3-17a4" siclog@

        f_EUTRA_CloseUE_TestLoopModeGH(eutra_Cell1, tsc_UE_TestLoopMode_TypeG, tsc_TestLoopModeGH_UplinkMode_NAS, 2, 0);

        //@siclog "Step 17a5" siclog@

        //SS transmits an ESM DATA TRANSPORT message containing 320 Octets of downlink user data.

        SRB.send(cas_SRB_NasPdu_REQ(eutra_Cell1,

                                tsc_SRB2,

                                cs_TimingInfo_Now,

                                cs_NAS_Request(tsc_SHT_IntegrityProtected_Ciphered,

                                               cs_ESM_DATA_TRANSPORT (tsc_EpsDefaultBearerId, v_PTI, cs_UserDataContainer(v_DataStep17a5)))));

        v_ReleaseAssistanceInd := cr_ReleaseAssistanceInd ('00'B) ifpresent;

        for (i:=0; i==1; i:=i+1) {  //Steps 17a6-17a8 are repeated  2 times.

            //@siclog "Step 17a6" siclog@

            //SS starts timer 1 min APN Rate control

            t_PLMNRateControl.start( f_EUTRA_SetTimerToleranceMax (eutra_Cell1, nasTimer, 60.0));

            //@siclog "Step 17a7" siclog@

            //Check: Does the UE send an ESM DATA TRANSPORT message respecting the APN set in step 17a1: Max size of message 256 Octets?

            SRB.receive(car_SRB2_NasPdu_IND(eutra_Cell1,

                                         cr_NAS_Indication(tsc_SHT_IntegrityProtected_Ciphered,

                                                           cr_ESM_DATA_TRANSPORT(tsc_EpsDefaultBearerId, cr_UserDataContainer(int2oct(lengthof(v_DataStep17a5),2),v_DataStep17a5), v_ReleaseAssistanceInd))));

            //@siclog "Step 17a8" siclog@

            //Wait until timer 1 min APN Rate control expires.

            t_PLMNRateControl.timeout;

            v_ReleaseAssistanceInd := cr_ReleaseAssistanceInd (('00'B, '01'B)) ifpresent;

        }

    }

    //Steps 18a1-18a6 describe behaviour that depends on UE capabilities; the "lower case letter" identifies a step sequence that take place depending on whether the UE supports MTU parameters

    if (pc_NonIP_Link_MTU_Parameter or pc_IPv4_Link_MTU_Parameter) {

        if (pc_IPv4_Link_MTU_Parameter) {

            v_ProtocolContainerIPv4 := cs_ProtocolContainer ('0010'O, int2oct(128, 2));

            v_ProtocolContainerList[v_PCOCount] := v_ProtocolContainerIPv4;

            v_PCOCount := v_PCOCount + 1 ;

        }

        if (pc_NonIP_Link_MTU_Parameter) {

            v_ProtocolContainerNonIP := cs_ProtocolContainer ('0015'O,  int2oct(128, 2));

            v_ProtocolContainerList[v_PCOCount] := v_ProtocolContainerIPv4;

            v_PCOCount := v_PCOCount + 1 ;

        }

        if (pc_APN_RateControl) {

            v_ProtocolContainerAPN := cs_ProtocolContainer ('0016'O, '00'O & 'FFFFFF'O);

            v_ProtocolContainerList[v_PCOCount] := v_ProtocolContainerIPv4;

        }

        v_NewPco := f_NAS_ExtdProtocolConfigOptionsTX (v_ProtocolContainerList);

        //@siclog "Step 18a1" siclog@

        //The SS sends a MODIFY EPS BEARER CONTEXT REQUEST message to modify APN, MTU and PLMN rate controls.

        SRB.send(cas_SRB_NasPdu_REQ(eutra_Cell1,

                                    tsc_SRB2,

                                    cs_TimingInfo_Now,

                                    cs_NAS_Request(tsc_SHT_IntegrityProtected_Ciphered,

                                                   cs_ModifyEPSBearerCxtReq_AllParams(tsc_EpsDefaultBearerId,

                                                                                      v_PTI,

                                                                                      omit,

                                                                                      omit,

                                                                                      omit,

                                                                                      omit,

                                                                                      omit,

                                                                                      f_Get_508_PFI(),

                                                                                      omit,

                                                                                      omit,

                                                                                      omit,

                                                                                      omit,

                                                                                      v_NewPco))));

       //@siclog "Step 18a2" siclog@

        //Check: Does the UE transmit a MODIFY EPS BEARER CONTEXT ACCEPT message?

        SRB.receive(car_SRB2_NasPdu_IND(eutra_Cell1,

                                    cr_NAS_Indication(tsc_SHT_IntegrityProtected_Ciphered,

                                                      cr_508_ModifyEPSBearerCxtAccept(tsc_EpsDedicatedBearerId))));

        f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Step 18a2");

        //@siclog "Steps 18a3-18a4" siclog@

        f_EUTRA_CloseUE_TestLoopModeGH(eutra_Cell1, tsc_UE_TestLoopMode_TypeG, tsc_TestLoopModeGH_UplinkMode_NAS, 1, 0);

        //@siclog "Step 18a5" siclog@

        //SS transmits an ESM DATA TRANSPORT message containing 240 Octets of downlink user data.

        SRB.send(cas_SRB_NasPdu_REQ(eutra_Cell1,

                                    tsc_SRB2,

                                    cs_TimingInfo_Now,

                                    cs_NAS_Request(tsc_SHT_IntegrityProtected_Ciphered,

                                                   cs_ESM_DATA_TRANSPORT (tsc_EpsDefaultBearerId, v_PTI, cs_UserDataContainer(f_GetN_OctetsFromPRBS(1, 240))))));

        alt {

            //@siclog "Step 18a6" siclog@

            [] SRB.receive(car_SRB2_NasPdu_IND(eutra_Cell1,

                                    cr_NAS_Indication(tsc_SHT_IntegrityProtected_Ciphered,

                                                      cr_ESM_DATA_TRANSPORT(tsc_EpsDefaultBearerId, cr_UserDataContainer(?,?))))) -> value v_ReceivedAsp

            {

                if (oct2int(v_ReceivedAsp.Signalling.Nas[0].Pdu.Msg.eSM_DATA_TRANSPORT.userDatacontainer.iel) <= 128) {

                    v_DataLengthReceived := v_DataLengthReceived + oct2int(v_ReceivedAsp.Signalling.Nas[0].Pdu.Msg.eSM_DATA_TRANSPORT.userDatacontainer.iel);

                } else {

                    f_EUTRA_SetVerdictFailOrInconc(__FILE__, __LINE__, "Step18a6");

                }

                if (v_DataLengthReceived < 240) { //All data not yet received

                    repeat;

                } else if (v_DataLengthReceived == 240) {

                    f_EUTRA_PreliminaryPass (__FILE__, __LINE__, "Step18a6");

                } else {

                    f_EUTRA_SetVerdictFailOrInconc(__FILE__, __LINE__, "Step18a6");

                }

            }

        }

    }

    //@siclog "Steps 19-20" siclog@

    //Deactivate Test mode

    f_EUTRA_Deactivate_TestMode(eutra_Cell1);

    //@siclog "Step 21" siclog@

    //The SS transmits an RRCConnectionRelease message.

    f_EUTRA_RRC_ConnectionRelease(eutra_Cell1);

    f_EUTRA_TestBody_Set(false);

    f_EUTRA_Postamble(eutra_Cell1, E1_IDLE, EPS_Only);

   }
After  change

  function f_TC_23_1_1_EUTRA() runs on EUTRA_PTC

  {/* CIoT / Control Plane MO and MT IP and non-IP Data Transfer / Serving PLMN Rate Control / APN Rate Control */

    var template (value) octetstring v_DataStep3_11 := 'F0F0F0'O;

    var template (value) octetstring v_DataStep17a5 := f_GetN_OctetsFromPRBS(1, 10);

    var EUTRA_SecurityParams_Type v_AuthParams;

    var ProcedureTransactionIdentifier v_PTI := tsc_PTI_Unassigned;

    var integer v_Count := 0;

    var integer v_PCOCount := 0;

    var integer v_DataLengthReceived := 0;

    var integer i;

    var template (value) ProtocolContainer v_ProtocolContainerAPN;

    var template (value) ExtdProtocolConfigOptions v_NewPco;

    var template (value) ProtocolContainer v_ProtocolContainerIPv4;

    var template (value) ProtocolContainer v_ProtocolContainerNonIP;

    var template (value) NAS_ExtdProtocolConfigOptions_Type v_ProtocolContainerList := {};

    var SRB_COMMON_IND v_ReceivedAsp;

    var template ReleaseAssistanceInd v_ReleaseAssistanceInd := cr_ReleaseAssistanceInd ('00'B) ifpresent;

    var template (value) APNRateControlParams v_APN_RateControl;

    timer t_LoopDelay;

    timer t_PLMNRateControl;

    //Initialise all cells, security and mobile parameters

    f_EUTRA_Init(c1);

    //Create and configure Cell 1

    f_EUTRA_CellConfig_Def (eutra_Cell1);

    //UE is in State 3A-CP, Control Plane CioT connection request, UE Test Mode Activated  according to TS 36.508

    //Settings in ACTIVATE DEFAULT EPS BEARER CONTEXT REQUEST, in preamble,  to check PLMN rate control => done in f_EUTRA_RRCConnectionReconfigurationWithAttachAcceptAllParams unsing attribute CIOT_CP_PLMN

    f_EUTRA_Preamble(eutra_Cell1, STATE2A_TESTLOOP_ModeG);

    f_EUTRA_RbEst_CP(eutra_Cell1);
    f_EUTRA_TestBody_Set(true);

    //@siclog "Steps 1 - 2" siclog@

    //The SS transmits a CLOSE UE TEST LOOP message to close the UE test loop mode for user data transfer (12 transmissions; 60 sec delay).

    //The UE transmits a CLOSE UE TEST LOOP COMPLETE message to confirm that loopback is activated.

    f_EUTRA_CloseUE_TestLoopModeGH(eutra_Cell1, tsc_UE_TestLoopMode_TypeG, tsc_TestLoopModeGH_UplinkMode_NAS, 12, 60);

    t_LoopDelay.start( f_EUTRA_SetTimerToleranceMin (eutra_Cell1, nonProtocolTimer, 60.0));

    //@siclog "Step 3" siclog@

    //SS transmits an ESM DATA TRANSPORT message containing downlink user data of 3 octets

    v_AuthParams := f_EUTRA_Security_Get();

    SRB.send(cas_SRB_NasPdu_REQ(eutra_Cell1,

                                tsc_SRB1,        //Anritsu TTCN CR requried 23.1.1 - SRB2 not configured after RRC connection release- use SRB1

                                cs_TimingInfo_Now,

                                cs_NAS_Request(tsc_SHT_IntegrityProtected_Ciphered,

                                               cs_ESM_DATA_TRANSPORT (tsc_EpsDefaultBearerId, v_PTI, cs_UserDataContainer(v_DataStep3_11)))));

    //@siclog "Step 4" siclog@

    //The SS transmits an RRCConnectionRelease message.

    f_EUTRA_RRC_ConnectionRelease ( eutra_Cell1);

    //@siclog "Step 5" siclog@

    //Wait for the time set in the CLOSE UE TEST LOOP to expire. => 60 seconds

    t_LoopDelay.timeout;

    //@siclog "Step 6" siclog@

    //SS starts timer 6 min PLMN Rate control

    t_PLMNRateControl.start( f_EUTRA_SetTimerToleranceMax (eutra_Cell1, nasTimer, 360.0));

    //@siclog "Step 7" siclog@

    //Check: Does the UE transmit an RRCConnectionRequest message?

    f_EUTRA_RRC_ConnectionRequest_Def(eutra_Cell1);

    //@siclog "Step 8" siclog@

    //SS transmits an RRCConnectionSetup message.

    f_EUTRA_RRC_ConnectionSetup_Def(eutra_Cell1);

    //@siclog "Step 9" siclog@

    //The UE transmits an RRCConnectionSetupComplete message to confirm the successful completion of the connection establishment and to initiate the session management procedure by including the CONTROL PLANE SERVICE REQUEST message.

    f_EUTRA_RRCConnectionSetupComplete_Def(eutra_Cell1,

                                           cr_508_RRCConnectionSetupComplete(tsc_RRC_TI_Def, ?),

                                           cr_NAS_IndicationWithPiggybacking(

                                                tsc_SHT_IntegrityProtected_PartiallyCiphered,

                                                cr_CONTROL_PLANE_SERVICE_REQUEST(v_AuthParams.KSIasme),

                                                cr_ESM_DATA_TRANSPORT(tsc_EpsDefaultBearerId, cr_UserDataContainer(int2oct(lengthof(v_DataStep3_11), 2),v_DataStep3_11))

                                                ));

    f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Step 9");

    //@siclog "Step 10" siclog@

    //The SS transmits a SERVICE ACCEPT message.

     SRB.send(cas_SRB1_NasPdu_REQ(eutra_Cell1, cs_TimingInfo_Now, cs_NAS_Request(tsc_SHT_IntegrityProtected_Ciphered, cs_SERVICE_ACCEPT)));
    //@siclog "Step 11" siclog@

    //Check: Does the UE send an ESM DATA TRANSPORT message containing the same user data sent by the SS in step 3?

    SRB.receive(car_SRB1_NasPdu_IND(eutra_Cell1, 
                                    cr_NAS_Indication(tsc_SHT_IntegrityProtected_Ciphered,

                                                      cr_ESM_DATA_TRANSPORT(tsc_EpsDefaultBearerId, cr_UserDataContainer(int2oct(lengthof(v_DataStep3_11),2),v_DataStep3_11)))));

    v_Count := 2;

    f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Step 11");

    //@siclog "Steps 12-16" siclog@

    alt {

      [] SRB.receive(car_SRB1_NasPdu_IND(eutra_Cell1,
                                         cr_NAS_Indication(tsc_SHT_IntegrityProtected_Ciphered,

                                                           cr_ESM_DATA_TRANSPORT(tsc_EpsDefaultBearerId, cr_UserDataContainer(int2oct(lengthof(v_DataStep3_11),2),v_DataStep3_11), v_ReleaseAssistanceInd))))

        {

          v_Count := v_Count +1;

          if (v_Count == 3) {

            f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Step 12");

            repeat;

          } else if (v_Count == 12) {

            f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Step 15");

            //@siclog "Step 16" siclog@

            // SS stops timer 6 min PLMN Rate control

            t_PLMNRateControl.stop;

          } else if (v_Count == 11){

            v_ReleaseAssistanceInd := cr_ReleaseAssistanceInd (('00'B, '01'B)) ifpresent;

            repeat; 

          }else {

            repeat;

          }

        }

      [v_Count == 10] t_PLMNRateControl.timeout

        {

           t_PLMNRateControl.start( f_EUTRA_SetTimerToleranceMax (eutra_Cell1, nasTimer, 360.0));

           repeat;

        }

    }

    //Steps 17a1 to 17a8 describe behaviour that depends on UE capabilities; the "lower case letter" identifies a step sequence that take place depending on whether the UE supports APN control

    if (pc_APN_RateControl) {

        //@siclog "Step 17a1" siclog@

        // The SS sends a MODIFY EPS BEARER CONTEXT REQUEST message of to modify APN, MTU and PLMN rate controls.

        v_APN_RateControl := cs_APNRateControlParams ('001'B, int2oct(1, 3));

        v_ProtocolContainerAPN := cs_ProtocolContainer ('0016'O, bit2oct(encvalue(v_APN_RateControl)));

        v_NewPco := f_NAS_ExtdProtocolConfigOptionsTX ({v_ProtocolContainerAPN});

        SRB.send(cas_SRB1_NasPdu_REQ(eutra_Cell1,
                                     cs_TimingInfo_Now,

                                     cs_NAS_Request(tsc_SHT_IntegrityProtected_Ciphered,

                                                    cs_ModifyEPSBearerCxtReq_AllParams(tsc_EpsDefaultBearerId,

                                                                                       v_PTI,

                                                                                       omit,

                                                                                       omit,

                                                                                       omit,

                                                                                       omit,

                                                                                       omit,

                                                                                       f_Get_508_PFI(),

                                                                                       omit,

                                                                                       omit,

                                                                                       omit,

                                                                                       omit,

                                                                                       v_NewPco))));

        //@siclog "Step 17a2" siclog@

        //Check: Does the UE transmit a MODIFY EPS BEARER CONTEXT ACCEPT message?

        SRB.receive(car_SRB1_NasPdu_IND(eutra_Cell1,
                                    cr_NAS_Indication(tsc_SHT_IntegrityProtected_Ciphered,

                                                      cr_508_ModifyEPSBearerCxtAccept(tsc_EpsDedicatedBearerId))));

        f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Step 17a2");

        //@siclog "Steps 17a3-17a4" siclog@

        f_EUTRA_CloseUE_TestLoopModeGH(eutra_Cell1, tsc_UE_TestLoopMode_TypeG, tsc_TestLoopModeGH_UplinkMode_NAS, 2, 0);

        //@siclog "Step 17a5" siclog@

        //SS transmits an ESM DATA TRANSPORT message containing 320 Octets of downlink user data.

        SRB.send(cas_SRB_NasPdu_REQ(eutra_Cell1,

                                tsc_SRB1,
                                cs_TimingInfo_Now,

                                cs_NAS_Request(tsc_SHT_IntegrityProtected_Ciphered,

                                               cs_ESM_DATA_TRANSPORT (tsc_EpsDefaultBearerId, v_PTI, cs_UserDataContainer(v_DataStep17a5)))));

        v_ReleaseAssistanceInd := cr_ReleaseAssistanceInd ('00'B) ifpresent;

        for (i:=0; i==1; i:=i+1) {  //Steps 17a6-17a8 are repeated  2 times.

            //@siclog "Step 17a6" siclog@

            //SS starts timer 1 min APN Rate control

            t_PLMNRateControl.start( f_EUTRA_SetTimerToleranceMax (eutra_Cell1, nasTimer, 60.0));

            //@siclog "Step 17a7" siclog@

            //Check: Does the UE send an ESM DATA TRANSPORT message respecting the APN set in step 17a1: Max size of message 256 Octets?

            SRB.receive(car_SRB1_NasPdu_IND(eutra_Cell1,
                                         cr_NAS_Indication(tsc_SHT_IntegrityProtected_Ciphered,

                                                           cr_ESM_DATA_TRANSPORT(tsc_EpsDefaultBearerId, cr_UserDataContainer(int2oct(lengthof(v_DataStep17a5),2),v_DataStep17a5), v_ReleaseAssistanceInd))));

            //@siclog "Step 17a8" siclog@

            //Wait until timer 1 min APN Rate control expires.

            t_PLMNRateControl.timeout;

            v_ReleaseAssistanceInd := cr_ReleaseAssistanceInd (('00'B, '01'B)) ifpresent;

        }

    }

    //Steps 18a1-18a6 describe behaviour that depends on UE capabilities; the "lower case letter" identifies a step sequence that take place depending on whether the UE supports MTU parameters

    if (pc_NonIP_Link_MTU_Parameter or pc_IPv4_Link_MTU_Parameter) {

        if (pc_IPv4_Link_MTU_Parameter) {

            v_ProtocolContainerIPv4 := cs_ProtocolContainer ('0010'O, int2oct(128, 2));

            v_ProtocolContainerList[v_PCOCount] := v_ProtocolContainerIPv4;

            v_PCOCount := v_PCOCount + 1 ;

        }

        if (pc_NonIP_Link_MTU_Parameter) {

            v_ProtocolContainerNonIP := cs_ProtocolContainer ('0015'O,  int2oct(128, 2));

            v_ProtocolContainerList[v_PCOCount] := v_ProtocolContainerIPv4;

            v_PCOCount := v_PCOCount + 1 ;

        }

        if (pc_APN_RateControl) {

            v_ProtocolContainerAPN := cs_ProtocolContainer ('0016'O, '00'O & 'FFFFFF'O);

            v_ProtocolContainerList[v_PCOCount] := v_ProtocolContainerIPv4;

        }

        v_NewPco := f_NAS_ExtdProtocolConfigOptionsTX (v_ProtocolContainerList);

        //@siclog "Step 18a1" siclog@

        //The SS sends a MODIFY EPS BEARER CONTEXT REQUEST message to modify APN, MTU and PLMN rate controls.

        SRB.send(cas_SRB_NasPdu_REQ(eutra_Cell1,

                                    tsc_SRB1,
                                    cs_TimingInfo_Now,

                                    cs_NAS_Request(tsc_SHT_IntegrityProtected_Ciphered,

                                                   cs_ModifyEPSBearerCxtReq_AllParams(tsc_EpsDefaultBearerId,

                                                                                      v_PTI,

                                                                                      omit,

                                                                                      omit,

                                                                                      omit,

                                                                                      omit,

                                                                                      omit,

                                                                                      f_Get_508_PFI(),

                                                                                      omit,

                                                                                      omit,

                                                                                      omit,

                                                                                      omit,

                                                                                      v_NewPco))));

       //@siclog "Step 18a2" siclog@

        //Check: Does the UE transmit a MODIFY EPS BEARER CONTEXT ACCEPT message?

        SRB.receive(car_SRB1_NasPdu_IND(eutra_Cell1,
                                    cr_NAS_Indication(tsc_SHT_IntegrityProtected_Ciphered,

                                                      cr_508_ModifyEPSBearerCxtAccept(tsc_EpsDedicatedBearerId))));

        f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Step 18a2");

        //@siclog "Steps 18a3-18a4" siclog@

        f_EUTRA_CloseUE_TestLoopModeGH(eutra_Cell1, tsc_UE_TestLoopMode_TypeG, tsc_TestLoopModeGH_UplinkMode_NAS, 1, 0);

        //@siclog "Step 18a5" siclog@

        //SS transmits an ESM DATA TRANSPORT message containing 240 Octets of downlink user data.

        SRB.send(cas_SRB_NasPdu_REQ(eutra_Cell1,

                                    tsc_SRB1,
                                    cs_TimingInfo_Now,

                                    cs_NAS_Request(tsc_SHT_IntegrityProtected_Ciphered,

                                                   cs_ESM_DATA_TRANSPORT (tsc_EpsDefaultBearerId, v_PTI, cs_UserDataContainer(f_GetN_OctetsFromPRBS(1, 240))))));

        alt {

            //@siclog "Step 18a6" siclog@

            [] SRB.receive(car_SRB1_NasPdu_IND(eutra_Cell1,
                                    cr_NAS_Indication(tsc_SHT_IntegrityProtected_Ciphered,

                                                      cr_ESM_DATA_TRANSPORT(tsc_EpsDefaultBearerId, cr_UserDataContainer(?,?))))) -> value v_ReceivedAsp

            {

                if (oct2int(v_ReceivedAsp.Signalling.Nas[0].Pdu.Msg.eSM_DATA_TRANSPORT.userDatacontainer.iel) <= 128) {

                    v_DataLengthReceived := v_DataLengthReceived + oct2int(v_ReceivedAsp.Signalling.Nas[0].Pdu.Msg.eSM_DATA_TRANSPORT.userDatacontainer.iel);

                } else {

                    f_EUTRA_SetVerdictFailOrInconc(__FILE__, __LINE__, "Step18a6");

                }

                if (v_DataLengthReceived < 240) { //All data not yet received

                    repeat;

                } else if (v_DataLengthReceived == 240) {

                    f_EUTRA_PreliminaryPass (__FILE__, __LINE__, "Step18a6");

                } else {

                    f_EUTRA_SetVerdictFailOrInconc(__FILE__, __LINE__, "Step18a6");

                }

            }

        }

    }

    //@siclog "Steps 19-20" siclog@

    //Deactivate Test mode

    f_EUTRA_Deactivate_TestMode_SRB1(eutra_Cell1);
    //@siclog "Step 21" siclog@

    //The SS transmits an RRCConnectionRelease message.

    f_EUTRA_RRC_ConnectionRelease(eutra_Cell1);

    f_EUTRA_TestBody_Set(false);

    f_EUTRA_Postamble(eutra_Cell1, E1_IDLE);
   }
5. Execution Log Files

5.1 Spreadtrum UIX8910
The Spreadtrum UIX8910 UE passed this CIoT TC in LTE FDD band 8 on Anritsu Protocol Conformance Test System ME7834LA.The documentation below is enclosed as evidence of the successful test case run [1]:

· Test Case Execution log file

(Note: PICS/PIXIT settings are captured at the beginning of the TLI log)


In the log file (in .txt / -html format) the complete test case execution can be seen. All TLI events are presented and message contents are fully decoded and can be verified. Preliminary verdicts and the final test case verdict can be seen in the log file.

